Open population models I: non-spatial models



OPEN POPULATION MODELS

- Open populations experience recruitment or mortality, emigration or immigration.
- N changes across time (seasons, years, or other), t=1,2,..,T
  N[t] = population size in year t
- We want to estimate N[t] from a model that is jointly specified for all the data. i.e., “multi-year model”

OPEN POPULATION MODELS

- **Two approaches to modeling open systems:**
  - **Approach 1**: Non-dynamic model. Model the period-specific N[t] parameters as independent parameters. ("implicit dynamics"?)
    - Years are regarded as independent strata.
    - N[t] are independent across years.
    - These approaches ignore individual identity across years and treat recaptures of the same individual as new individuals.
    - The approach is therefore less efficient for estimating N[t] but often not by much.
  - **Approach 2**: Dynamic model. ("explicit dynamics") Preserves individual identity – more efficient use of data. Model transitions of individuals from one year to the next.
    - N[t] are not independent across years.
    - Allows estimation of dynamics parameters.
    - JS and CJS type models

NON-DYNAMIC OPEN MODELS

Two approaches to implementation:

- Year as an individual class variable:

  ```
  year[i] ~ dcat(probs[year])
  ```

  - This allocates the uncaptured individuals into years
  - Regards individuals as independent across years (recaptures are new individuals)
  - We already did this on Day 1

- T-fold data augmentation. Analyze T data sets, each one augmented individually with DA parameter psi[t]. Operationally: Input the data as a 3-d array "individual x replicate x year"
  - Also regards individuals as independent across years

- These are practically equivalent if dim(probs[year]) = T-1

OPEN POPULATIONS

- Both classes of models are easily ‘spatialized’. i.e., once we have the ordinary non-spatial model described in BUGS, then we can easily add the latent activity centers s[i] with a few lines of code

- Therefore we will spend a fair amount of time talking about the non-spatial versions of these models.

HOW TO DO THIS IN BUGS

T-fold data augmentation:

Create T data sets, augmented to the same dimension M (for convenience)

- 3-d array data structure: Store the data in an M x K x T array with elements y[i,k,t] = 1 if captured during occasion k year t, etc..
- 2-d version based on summary statistics. If there are no effects that vary in the k-dimension (behavioral response, occasion-specific parameters) then we can summarize the data to y[i,t] = total number of captures out of K

T-FOLD DATA AUGMENTATION WITH THE 3-D ARRAY

```
for(t in 1:5){
  N[t] <- sum(z[ ,t])
  psi[t] ~ dunif(0,1)
  p[t] ~ dunif(0,1)
}
for(t in 1:5){
 for(i in 1:M){
  z[i,t] ~ dbern(psi[t])
   for(k in 1:K){
      mu[i,k,t]<- z[i,t]*p[t]*(1-isdead[i,k,t])
      Yarr[i,k,t] ~ dbern( mu[i,k,t] )
     }
    }
 }   }
```

T-FOLD DATA AUGMENTATION: DATA SUMMARIZED TO 2-D ARRAY

```
model {

for(t in 1:5){
  N[t] <- sum(z[ ,t])
  psi[t] ~ dunif(0,1)
  p[t] ~ dunif(0,1)
}

for(t in 1:5){
 for(i in 1:M){
  z[i,t] ~ dbern(psi[t])
    mu[i,t]<- z[i,t]*p[t]
   Y2d[i,t] ~ dbin( mu[i,t], Kmat[i,t] )  # Kmat[i,t] instead of isdead
   }
   }
   }
```

T-FOLD DATA AUGMENTATION

- R work session

TOWARD A DYNAMIC MODELS

- These approaches do not preserve individual ID across years

- It is possible to extend the "year as a class variable" specification to include individual ID but it is inelegant and untidy.

- The 3-d array or "stacked" T-fold DA models can be extended to preserve individual ID

- Ignoring individual identity across years, treating recaptures as new individuals, disregards a small bit of information and thus we're making inefficient use of data, not getting the most efficient estimates (in a variance sense), and also not able to estimate survival and recruitment

PRESERVING INDIVIDUAL IDENTITY IN OPEN MODELS

Population model with survival and recruitment is exactly a restricted occupancy model (Ch. 10 Royle and Dorazio 2008). Recruitment = colonization, 1-extinction = survival. The restriction is: once an animal dies it can't be recruited again.

If an individual is currently alive:

```
z[i,t] ~ Bern( phi*z[i,t-1] )
```

If an individual is currently not yet recruited prior to t

```
z[i,t] ~ Bern( gamma*z[i,t-1] )
```

To implement this model we need to keep track of whether an animal was ever alive, so that it cannot be re-recruited….

We take code directly from Kery and Schaub (2012, Ch. 10).

BUGS CODE FOR JOLLY-SEBER MODEL

```
# Priors and constraints
for (i in 1:M){
   for (t in 1:( T-1 )){
       phi[i,t] <- mean.phi
     } #t
   for (t in 1:T){
       p[i,t] <- mean.p
     } #t
   } #i
mean.phi ~ dunif(0, 1)
mean.p ~ dunif(0, 1)
for (t in 1:T){
   gamma[t] ~ dunif(0, 1)      # year-specific colonization probability
   N[t] <- sum(z[1:M,t])       # Actual population size
   } #t
# Observation model
for (i in 1:M){
for(t in 1:T){
   mu1[i,t] <- z[i,t] * p[i,t]
   y[i,t] ~ dbin(mu1[i,t],Kmat[i,t])
}
}
## Process model
for(i in 1:M){
   # Initial state
   z[i,1] ~ dbern(gamma[1])
   # Subsequent occasions
   for (t in 2:T){
      q[i,t-1] <- 1-z[i,t-1]                        # Availability for recruitment
      available[i,t-1]<- prod(q[i,1:(t-1)])
      mu2[i,t] <- phi[i,t-1] * z[i,t-1] + gamma[t] * available[i,t-1]
      z[i,t] ~ dbern(mu2[i,t])
       } #t
   } #i
```

Constant phi and p here but could vary by i and t easily

JS MODEL: RESTRICTED OCCUPANCY MODEL

- Useful derived parameters:
  - N[t] = number of occupied sites in period t
  - B[t] = number of newly occupied sites in period t

Open population models II: SCR models



OPEN SCR MODELS

Two types of "open" dynamics we can model:

1. Population dynamics: individuals are entering and leaving the population through recruitment and mortality

2. Spatial dynamics – dynamic activity centers (dispersal, migration, transience)

Open SCR models can have one or both things happening

CLASSES OF "OPEN" SCR MODELS

- Closed SCR model with dynamic activity centers
- Implicit dynamics ("N random"), static activity centers
- Explicit dynamics model (Jolly-Seber), static activity centers
- Explicit dynamics model with dynamic activity centers

"open population models" should be "dynamic population models" and we have different types of dynamics acting on different components of the model.

5 (OR 6) CLASSES OF "OPEN" SCR MODELS

SPATIAL DYNAMICS

POPULATION DYNAMICS	s[i] static	s[i] dynamic
None	Basic SCR model (not open)	SCR + dispersal or transience
Random N[t]	"year as a class variable" s[i,t] ~ random	[no individual ID]
Partial dynamics	Cormack-Jolly-Seber	+ s[i,t]
Fully dynamic	Jolly-Seber model	+ s[i,t]

“OPEN” SCR I: CLOSED SCR MODEL WITH SPATIAL DYNAMICS

- Ordinary SCR model with transience or dispersal
- Accounts for non-closure due to animals moving away from the area of the state-space where they were catchable.
- Accounts for correlated space usage by individuals
- Modifications to the model:
  - s[i,**k**] instead of s[i]
  - Model for s[i,**k**]

Different “sigma” here

Plausible models:

1. s[i,k] ~ Uniform(S) # This model is not estimable
2. s[i,k] ~ normal(mu[i], sigma^2) # mu[i]: meta-homerange center
3. s[i,k] ~ normal(s[i,k-1], sigma^2) # Markovian transience

CLOSED SCR MODEL WITH SPATIAL DYNAMICS

- For the Markovian dispersal or transience models we need to increase the size of the state-space to allow sufficient area for movement (relative to size of movement scale parameter)

- Also have to preserve density of the state-space – individuals can't leave. It is easiest to truncate the movement/dispersal model to restrict s[i,k]

- **Demonstration using the Fort Drum black bear data**

OPEN SCR II: IMPLICIT DYNAMICS, RANDOM ACTIVITY CENTERS

- Here we spatialize that model in two ways:
  - 1. Including static activity centers s[i] – suggesting that the home ranges are stationary over time
  - 2. by including activity centers for each individual and year, s[i,t]
  - “secr” can fit model 2 using the multi-session approaches.
- These models accommodate implicit dynamics and possible changes in activity centers

OPEN SCR III: SPATIAL JOLLY-SEBER TYPE MODEL

- Variations of this model:
  1. JS model with static activity centers
  2. JS model with random activity centers
  3. JS model with dynamic activity centers

SUMMARY

- Many variations of “non-closure” in SCR models:
  - Spatially dynamic activity centers
  - Population dynamics (survival and recruitment)

- Can have spatial dynamics operating even in closed populations

- A benefit of using BUGS is all of these models are relatively easy to parameterize and analyse.

- In ‘secr’ you can only do limited things.